/*
CpGears v1.0

Copyright (c) 2010 Yves Riel http://www.flash-factor.com

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in
all copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN
THE SOFTWARE.
*/
package cpgears.movie.slide {
	import cpgears.helpers.StringUtil;
	import flash.utils.Dictionary;
	
	/**
	 * Enumeration class containing the Captivate Slide Types
	 * 
	 * @author Whyves
	 */
	public class SlideType{
		
		// TRICKY: The static dictionairy must be the first one declared.
		private static var mEnumeration:Dictionary = new Dictionary();
		
		/**
		 *  A normal slide
		 */
		public static const NORMAL:SlideType = new SlideType("Normal Slide");
		
		/**
		 *  A question slide
		 */
		public static const QUESTION:SlideType = new SlideType("Question Slide");

		/**
		 *  A random question slide
		 */
		public static const RANDOM_QUESTION:SlideType = new SlideType("Random Question Slide");
		
		private var mName:String = null;
		
		/**
		 * Creates a slide type.
		 * 
		 * @param	name The slide internal name.
		 */
		public function SlideType(name:String) {
			if (StringUtil.isNullOrEmpty(name)) throw new ArgumentError("'name' parameter cannot be null or empty.");
			mName = name;
			mEnumeration[name] = this;
		}

		/**
		 * Returns the slide type which internal name matches the paramter.
		 * 
		 * @param	typeName The internal name of the slide.
		 * @return The slide matching the internal name.
		 */
		public static function getSlideTypeFromName(typeName:String):SlideType {
			return (mEnumeration.hasOwnProperty(typeName)) ? mEnumeration[typeName] : new SlideType(typeName);
		}
		
		/**
		 * Returns the slide's internal name
		 */
		public function get value():String {
			return mName;
		}
	}
}